﻿* Encoding: UTF-8.

GET   FILE='[file location]/CCES20_Replication_Wilson and Davis.sav'.
DATASET NAME $DataSet WINDOW=FRONT.

******************************************************************************************************************************.
*********************** Data Coding ***********************************************************************************.
******************************************************************************************************************************.

*******************.
* Demographics.
********************.
* Age.
compute age = 2018 -  birthyr.
recode age (lo thru 29=1) (30 thru 49=2) (50 thru hi=3) into agecat3.
value labels agecat3 1 '18-29' 2 '30-49' 3 '50+'.
recode age (lo thru 29=1) (30 thru 39=2) (40 thru 49=3) (50 thru 59=4) (60 thru hi=5) into agecat5.
value labels agecat5 1 '18-29 (Born 1987-1998)' 2 '30-39 (Born 1977-1986)' 3 '40-49 (Born 1967-1976 )'
   4 '50-59 (Born 1957-1966)' 5 '60+ (Born 1956 or earlier)'.
recode birthyr (LO thru 1959=1) (1960 thru 1969=2) (1970 thru 1979=3) (1980 thru 1989=4) (1990 thru hi=5) into cohort.
value labels cohort 1 'Before 1960' 2 '1960 - 1969' 3 '1970 - 1979' 4 '1980 - 1989' 5 '1990 - 1998'.
recode age (lo thru 29=1) (30 thru 39=2) (40 thru 49=3) (50 thru 59=4) (60 thru hi=5) into age5.
value labels age5 1 '18-29' 2 '30-39' 3 '40-49' 4 '50-59' 5 '60+'.

* Sex.
compute male = gender=1.
compute female = gender=2.

* Educ.
recode educ (1 thru 4=0) (5,6=1) into college.
value labels college 1 '4 year degree' 0 'no 4 year degree'.
recode educ (1,2=1) (3,4=2) (5,6=3) into educ3.
value labels educ3 1 'HS or less' 2 'post-HS' 3 'College degree'.

* Race.
recode race (1=1) (1=1) (2=2) (3=3) (4=4) into race4.
value labels race4 1 'white' 2 'black' 3 'hispanic' 4 'asian'.
recode race (1=1) (1=1) (2=2) (3=3)  into race3.
value labels race3 1 'white' 2 'black' 3 'hispanic' .
recode race (1=0) (2 thru 8=1) into minority.
value labels minority 1 'Non-White' 0 'White'.
recode race (1, 3, 4=0) (2=1) (5,7,8=0) into black.
recode race (1, 2,4=0) (3=1) (5,7,8=0) into hispanic.
recode race (1,2,3=0) (4=1) (5,7,8=0) into asian.
recode race (1 thru 5=0) (6=1) (7,8=0) into multirace.
recode race (1 thru 4=0) (5,7,8=1) (6=0) into other.

recode race (1=1) (2=2) (3=3) (4=4) (5 thru 8=5) (98,99=SYSMIS) into race5.
value labels race5 1 'White' 2 'Black' 3 'Hispanic' 4 'Asian' 5 'Other'.
recode race5 (1=1) (2 thru 5=0) into Whites.
value labels whites 1 'White resp' 0 'all other (non-AfrAm)'.
compute asian = race5=4.
recode race4 (1=1) (2=2) into whtblk2.
value labels whtblk2 1 'Whites' 2 'African Americans'.
recode race4 (1=1) (2=0) into whtblk_dum.
value labels whtblk_dum 1 'Whites' 0 'African Americans'.
recode race4 (1=0) (2 thru 4=1) into minority.
value labels minority 1 'Non-White' 0 'White'.
recode race4 (3,4=1) (1,2=0) into xraceethn.
value labels xraceethn 1 'non Blk minorities' 0 'whts & blks'.
exe.

* Filter variables.
compute whitesample = race=1.
compute blacksample=race=2.
compute whtblksample=race=1 OR race=2.
exe.

* Ideology.
missing values ideo5 (6).
AUTORECODE ideo5 /into conservatism /print.
missing values ideo5 (6,8,9) .
recode ideo5 (1,2=1) (3=2) (4,5=3) into ideo3.
value labels ideo3 1 'liberal' 2 'moderate' 3 'conservative'.

* Party ID.
missing values pid7 (8,9).
AUTORECODE pid7 /into republicanism /print.
recode pid7 (1 thru 3=1) (4=2) (5 thru 7=3) (ELSE = SYSMIS) into party3.
value labels party3 1 'Democrat' 2 'Independent' 3 'Republican'.
compute democrat = party3=1.
compute independent = party3=2.
compute republican = party3=3.
recode pid7 (1,2=1) (3=2) (4=3) (5=4) (6,7=5) into party5.
value labels party5 1 'Dem/Str Dem' 2 'Lean Dem' 3 'Independent' 4 'Lean Rep' 5 'Rep/Str Rep'.

* Religion.
AUTORECODE  pew_religimp /into xreligion /desc /print.

* Family Income (Estimated).
missing values faminc_new (17, 97,98).
autorecode faminc_new /into incomefam .

*********** IMPUTE income **************************.
REGRESSION    /MISSING LISTWISE   /STATISTICS COEFF OUTS R ANOVA
  /CRITERIA=PIN(.05) POUT(.10)   /NOORIGIN    /DEPENDENT incomefam
  /METHOD=ENTER age educ male whites   .
missing values incomefam ().
do if (incomefam=17).
COMPUTE incomefam=1.89+(.005*age)+(.945*educ)+(1.273*male)+(.56*whites).
end if.
variable level incomefam (scale).


*incomefam.
RECODE  incomefam (MISSING=COPY) (LO THRU 3=1) (LO THRU 5=2) (LO THRU 7=3) (LO THRU 9=4) (LO THRU 
    HI=5) (ELSE=SYSMIS) INTO faminc5.
VARIABLE LABELS  faminc5 'Family income (Binned)'.
VALUE LABELS  faminc5 1 '1st ntile (<= 3)' 2 '2nd ntile (4 - 5)' 3 '3rd ntile (6 - 7)' 4 '4th '+
    'ntile (8 - 9)' 5 '5th ntile (10+)'.
VARIABLE LEVEL  faminc5 (ORDINAL).

* Deep South Residence.
compute southdeep = ANY(inputstate, 1, 12, 13, 22, 28, 45, 48 )=1.
value labels southdeep 1 'Deep South (7 states)' 0 'All others'.
exe.

**********************************.
* Racial resentment.
**********************************.
AUTORECODE    UND301_1 UND301_2 UND301_3 UND301_4 UND301_5 /into
WRTA_nodiff WRTA_resent WRTA_stop WRTA_unfair WRTA_excuse /desc /print.
COMPUTE WRTA = Mean.5(WRTA_nodiff, WRTA_resent, WRTA_stop, WRTA_unfair, WRTA_excuse).
exe.

* Visual Binning.
*WRTA.
filter by whitesample.
RECODE  WRTA (MISSING=COPY) (LO THRU 2=1) (LO THRU 3.6=2) (LO THRU HI=3) (ELSE=SYSMIS) INTO 
    WRTA_3cat.
VARIABLE LABELS  WRTA_3cat 'WRTA (Binned)'.
FORMATS  WRTA_3cat (F5.0).
VALUE LABELS  WRTA_3cat 1 'Low' 2 'Medium' 3 'High'.
VARIABLE LEVEL  WRTA_3cat (ORDINAL).
use all.
EXECUTE.

************************************.
* White Privilege and Perceived Racial Problems items.
************************************.

AUTORECODE CC20_440a  /into FIRE_whtpriv /desc /print.
AUTORECODE CC20_440a  /into FIRE_whtpriv_r  /print.
AUTORECODE  CC20_440b /into FIRE_rare /desc /print.

filter by whitesample.
exe.
freq fire_rare FIRE_whtpriv.
descriptives FIRE_rare FIRE_whtpriv.

USE ALL.
recode fire_rare (4,5=3) (3=2) (1,2=1) into fire_rare3.
variable labels FIRE_rare3 'Racial/Sexual agreement -- Racial problems in the U.S. are rare, isolated situations'.
recode FIRE_whtpriv (4,5=3) (3=2) (1,2=1) into fire_whtpriv3.
variable labels fire_whtpriv3 'White people in the U.S. have certain advantages because of the color of their skin'.
value labels FIRE_rare3 fire_whtpriv3 1 'Disagree' 2 'Neither' 3 'Agree'.
exe.

************************************.
* Feeling Thermometers (Racial Affect).
************************************.
use all.
AUTORECODE  UND420 UND421 UND422  /into   ft_blacks ft_latinos ft_whites  /print.
DESCRIPTIVES  ft_blacks  ft_whites ft_latinos   .

************************************.
* Resistance to change.
************************************.
AUTORECODE  UND301_6 UND301_7 UND301_8 UND301_9 UND301_10 UND301_11 UND301_12 UND301_13 UND301_14 UND301_15
 /INTO BJW1 SJT1 BJW2 SJT2 AUTH1 ENABLE1 ENABLE2 ENABLE3 ENABLE4 ENABLE5 /desc /print.
AUTORECODE  UND301_12  /INTO   ENABLE2_r /print.  

Compute BJW_scale2 = mean.4(bjw1, bjw2 ,sjt1, sjt2).
exe.

RECODE  BJW_scale2 (MISSING=COPY) (LO THRU 2.5=1) (LO THRU 3.75=2) (LO THRU HI=3) (ELSE=SYSMIS) INTO BJW_scale2_3cat.
VARIABLE LABELS  BJW_scale2_3cat 'BJW_scale2 (Binned)'.
FORMATS  BJW_scale2_3cat (F5.0).
VALUE LABELS  BJW_scale2_3cat 1 'Low' 2 'Medium' 3 'High'.
VARIABLE LEVEL  BJW_scale2_3cat (ORDINAL).

************************************.
* Antiracism Action Orientation.
************************************.
COMPUTE ENABLE_3itemR = MEAN.3(UND301_11,  UND301_13,  UND301_15).
exe.

******************************************************************************************************************************.
*********************** Begin Analysis *************************************************************************************.
******************************************************************************************************************************.

*****************************************.
* Table 1.
*****************************************.
FREQUENCIES VARIABLES=UND301_11  UND301_13  UND301_15
  /STATISTICS=STDDEV MEAN MEDIAN   /ORDER=ANALYSIS.

*****************************************.
* Table 2.
*****************************************.
use all.
MEANS TABLES=ENABLE_3itemR BY  race4  minority /CELLS=MEAN  STDDEs by minority  /STATISTICS ANOVA.
T-TEST GROUPS=minority(1 0)    /MISSING=ANALYSIS   /VARIABLES=ENABLE_3itemr
  /ES DISPLAY(TRUE)   /CRITERIA=CI(.95).

sort cases by minority.
split file by minority.
T-TEST GROUPS=race4(3)   /MISSING=ANALYSIS
  /VARIABLES=ENABLE_3itemR   /ES DISPLAY(TRUE)   /CRITERIA=CI(.95).
ONEWAY ENABLE_3itemR BY race4
  /STATISTICS DESCRIPTIVES   /MISSING ANALYSIS   /CRITERIA=CILEVEL(0.95)
  /POSTHOC=BONFERRONI ALPHA(0.05).
MEANS TABLES=ENABLE_3itemR BY  race4   /CELLS=MEAN  STDDEs  STDDEV COUNTV /STATISTICS ANOVA.
split file off.

MEANS TABLES=ENABLE_3itemR BY  age5   /CELLS=MEAN  STDDEV COUNTV /STATISTICS ANOVA.

T-TEST GROUPS=male(1 0)    /MISSING=ANALYSIS   /VARIABLES=ENABLE_3itemR
  /ES DISPLAY(TRUE)   /CRITERIA=CI(.95).
MEANS TABLES=ENABLE_3itemR BY  male   /CELLS=MEAN  STDDEs  STDDEV COUNTV /STATISTICS ANOVA.

T-TEST GROUPS=college(1 0)    /MISSING=ANALYSIS   /VARIABLES=ENABLE_3itemR
  /ES DISPLAY(TRUE)   /CRITERIA=CI(.95).
MEANS TABLES=ENABLE_3itemR BY  college   /CELLS=MEAN  STDDEs  STDDEV COUNTV /STATISTICS ANOVA.

MEANS TABLES=ENABLE_3itemR BY  ideo3   /CELLS=MEAN  STDDEV COUNt /STATISTICS ANOVA.

MEANS TABLES=ENABLE_3itemR BY  party3   /CELLS=MEAN  STDDEV COUNT /STATISTICS ANOVA.

T-TEST GROUPS=southdeep(1 0)    /MISSING=ANALYSIS   /VARIABLES=ENABLE_3itemR
  /ES DISPLAY(TRUE)   /CRITERIA=CI(.95).
MEANS TABLES=ENABLE_3itemR BY   southdeep   /CELLS=MEAN  STDDEV COUNT /STATISTICS ANOVA.

MEANS TABLES=ENABLE_3itemR BY  xreligion   /CELLS=MEAN  STDDEV COUNT/STATISTICS ANOVA.

MEANS TABLES=ENABLE_3itemR BY  faminc5   /CELLS=MEAN  STDDEV COUNT /STATISTICS ANOVA.
freq faminc_new .

filter by whitesample.
MEANS TABLES=ENABLE_3itemR BY  FIRE_whtpriv3   /CELLS=MEAN  STDDEV COUNT /STATISTICS ANOVA.
use all.

* Corrs for t-tests.
MEANS TABLES=ENABLE_3itemR BY   minority male college southdeep   /CELLS=MEAN  STDDEV COUNT /STATISTICS ANOVA.
use all.
exe.

*****************************************.
* Table 3 - Zero Order.
*****************************************.
CORR  fire_rare   ft_blacks   ft_latinos ft_whites black xraceethn  BJW_scale2 FIRE_rare WRTA WITH ENABLE_3itemR /print nosig.

*****************************************.
* Table 3 (also Table SM1 and SM7 (VIF estimates)) - Full Model Part Estimates - Model 1.
*****************************************.

Weight off.
REGRESSION  /MISSING LISTWISE   /STATISTICS COEFF OUTS R ANOVA CHANGE ZPP TOL /descriptives CORR SIg N
  /CRITERIA=PIN(.05) POUT(.10)   /NOORIGIN 
  /DEPENDENT ENABLE_3itemR
  /METHOD=ENTER   age female college  faminc5  xreligion southdeep democrat independent conservatism   ft_blacks  ft_latinos ft_whites black xraceethn  BJW_scale2  FIRE_rare .

*****************************************.
* Table 3 (also Table SM1) - Full Model (Whites and Racial Resentment) - Model 2. 
*****************************************.

Filter by whitesample .
REGRESSION  /MISSING LISTWISE   /STATISTICS COEFF OUTS R ANOVA CHANGE ZPP TOL /descriptives CORR SIg N
  /CRITERIA=PIN(.05) POUT(.10)   /NOORIGIN 
  /DEPENDENT ENABLE_3itemR
  /METHOD=ENTER   age female college  faminc5  xreligion southdeep democrat independent conservatism   ft_blacks  ft_latinos ft_whites  BJW_scale2  FIRE_rare WRTA.
use all.
exe.

****************************************************************************************.
***************** Figure 1 and Table SM2 *****************************************.
****************************************************************************************.
Filter by whitesample .
exe. 
UNIANOVA ENABLE_3itemR WITH age female college  faminc5 xreligion southdeep democrat independent conservatism ft_blacks ft_whites ft_latinos BJW_scale2 FIRE_rare WRTA
  /METHOD=SSTYPE(3)
  /INTERCEPT=INCLUDE
  /PRINT ETASQ PARAMETER DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /DESIGN=  age female college  faminc5  xreligion southdeep democrat independent conservatism  ft_blacks ft_latinos ft_whites   BJW_scale2  FIRE_rare WRTA
      BJW_scale2*wrta     
       /save pred(Y3hat).

GRAPH   /LINE(MULTIPLE)=MEAN(y3hat) BY  WRTA_3cat BY BJW_scale2_3cat  /INTERVAL CI(95.0).

****************************************************************************************.
************* Figure 2 and Tablea SM3a, SM3b, SM3c***************************.
****************************************************************************************.

* Figure 2a and 2b (Table SM3a).
sort cases by party3.
split file by party3.
UNIANOVA ENABLE_3itemR WITH age female college  faminc5 xreligion southdeep  conservatism  BJW_scale2 WRTA
  /METHOD=SSTYPE(3)
  /INTERCEPT=INCLUDE
  /PRINT ETASQ PARAMETER DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /DESIGN=  age female college  faminc5  xreligion southdeep conservatism  BJW_scale2 WRTA 
  /save pred(yparty)  .
split file off.
GRAPH   /LINE(MULTIPLE)=MEAN(  yparty) BY  BJW_scale2_3cat BY party3  /INTERVAL CI(95.0)  .
GRAPH   /LINE(MULTIPLE)=MEAN(  yparty) BY  WRTA_3cat BY party3   /INTERVAL CI(95.0)  .

* Figure 2c and 2d (Table SM3b).
sort cases by ideo3.
split file by ideo3.
UNIANOVA ENABLE_3itemR WITH age female college  faminc5 xreligion southdeep  democrat independent   BJW_scale2 WRTA
  /METHOD=SSTYPE(3)
  /INTERCEPT=INCLUDE
  /PRINT ETASQ PARAMETER DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /DESIGN=  age female college  faminc5 xreligion southdeep  democrat independent   BJW_scale2 WRTA   
    /save pred(yideo)  .
split file off.
GRAPH   /LINE(MULTIPLE)=MEAN(  yideo) BY  BJW_scale2_3cat BY ideo3  /INTERVAL CI(95.0)  .
GRAPH   /LINE(MULTIPLE)=MEAN(  yideo) BY  WRTA_3cat BY ideo3   /INTERVAL CI(95.0)  .

* Figure 2e and 2f (Table SM3c).
sort cases by  fire_whtpriv3.
split file by  fire_whtpriv3.
UNIANOVA ENABLE_3itemR WITH age female college  faminc5 xreligion southdeep  democrat independent conservatism  BJW_scale2 WRTA
  /METHOD=SSTYPE(3)
  /INTERCEPT=INCLUDE
  /PRINT ETASQ PARAMETER DESCRIPTIVE
  /CRITERIA=ALPHA(.05)
  /DESIGN=  age female college  faminc5  xreligion southdeep democrat independent conservatism  BJW_scale2 WRTA   
      /save pred(ypriv)  .
split file off.
GRAPH   /LINE(MULTIPLE)=MEAN(ypriv) BY  BJW_scale2_3cat BY fire_whtpriv3  /INTERVAL CI(95.0)  .
GRAPH   /LINE(MULTIPLE)=MEAN(ypriv) BY  WRTA_3cat BY fire_whtpriv3   /INTERVAL CI(95.0)  .

use all.
exe.

*********************************.
* FIgure SM1.
*********************************.
GRAPH   /HISTOGRAM(NORMAL)=ENABLE_3itemR.

*********************************.
* Table SM4 .
*********************************.
filter by whitesample.
DESCRIPTIVES UND301_11  UND301_13  UND301_15  WRTA_nodiff WRTA_resent WRTA_stop WRTA_unfair WRTA_excuse BJW1 BJW2 SJT1 SJT2  FIRE_rare /MISSING LISTWISE .
CORR UND301_11  UND301_13  UND301_15 WRTA_nodiff WRTA_resent WRTA_stop WRTA_unfair WRTA_excuse BJW1 BJW2 SJT1 SJT2  FIRE_rare /print nosig /missing listwise .
FACTOR   /VARIABLES UND301_11  UND301_13  UND301_15 WRTA_nodiff WRTA_resent WRTA_stop WRTA_unfair WRTA_excuse BJW1 BJW2 SJT1 SJT2  FIRE_rare 
  /MISSING LISTWISE 
  /ANALYSIS UND301_11  UND301_13  UND301_15  WRTA_nodiff WRTA_resent WRTA_stop WRTA_unfair WRTA_excuse BJW1 BJW2 SJT1 SJT2  FIRE_rare 
  /PRINT UNIVARIATE INITIAL KMO EXTRACTION ROTATION
  /CRITERIA MINEIGEN(1) ITERATE(25)   /EXTRACTION PAF   /CRITERIA ITERATE(25)
  /ROTATION VARIMAX   /METHOD=CORRELATION.


*********************************.
* Table SM5 (the SM5 items have the "ENABLE" label rather than the original data labels used in earlier analyses.
*********************************.
DESCRIPTIVES UND301_11 UND301_12 UND301_13 UND301_14 UND301_15 .
AUTORECODE   UND301_11 UND301_12 UND301_13 UND301_14 UND301_15
 /INTO  ENABLE1 ENABLE2 ENABLE3 ENABLE4 ENABLE5 /desc /print.

* reverse coding ENABLE2.
AUTORECODE  UND301_12  /INTO   ENABLE2_r /print.  

* Factor analysis and Alpha Reliability - All 5 items. 
FACTOR   /VARIABLES ENABLE1 ENABLE2 ENABLE3 ENABLE4 ENABLE5
  /MISSING LISTWISE 
  /ANALYSIS ENABLE1 ENABLE2 ENABLE3 ENABLE4 ENABLE5
  /PRINT UNIVARIATE INITIAL KMO EXTRACTION ROTATION
  /CRITERIA MINEIGEN(1) ITERATE(25)   /EXTRACTION PAF   /CRITERIA ITERATE(25)
  /ROTATION VARIMAX   /METHOD=CORRELATION.
 RELIABILITY  /VARIABLES= ENABLE1  ENABLE2 ENABLE3 ENABLE4 ENABLE5
 /SCALE('ALL VARIABLES') ALL   /MODEL=ALPHA   /STATISTICS=HOTELLING   /SUMMARY=TOTAL. 

* Factor analysis and Alpha Reliability - All 5 items with item #2 reverse coded. 
FACTOR   /VARIABLES ENABLE1 ENABLE2_r ENABLE3 ENABLE4 ENABLE5
  /MISSING LISTWISE 
  /ANALYSIS ENABLE1 ENABLE2_r ENABLE3 ENABLE4 ENABLE5
  /PRINT UNIVARIATE INITIAL KMO EXTRACTION ROTATION
  /CRITERIA MINEIGEN(1) ITERATE(25)   /EXTRACTION PAF   /CRITERIA ITERATE(25)
  /ROTATION VARIMAX   /METHOD=CORRELATION.
 RELIABILITY  /VARIABLES= ENABLE1  ENABLE2_r ENABLE3 ENABLE4 ENABLE5
 /SCALE('ALL VARIABLES') ALL   /MODEL=ALPHA   /STATISTICS=HOTELLING   /SUMMARY=TOTAL. 

* Factor analysis and Alpha Reliability - Final 3 items.
FACTOR   /VARIABLES ENABLE1 ENABLE3 ENABLE5
  /MISSING LISTWISE 
  /ANALYSIS ENABLE1  ENABLE3 ENABLE5
  /PRINT UNIVARIATE INITIAL KMO EXTRACTION ROTATION
  /CRITERIA MINEIGEN(1) ITERATE(25)   /EXTRACTION PAF   /CRITERIA ITERATE(25)
  /ROTATION VARIMAX   /METHOD=CORRELATION.
RELIABILITY  /VARIABLES= ENABLE1  ENABLE3  ENABLE5
  /SCALE('ALL VARIABLES') ALL   /MODEL=ALPHA   /STATISTICS=HOTELLING   /SUMMARY=TOTAL. 


*********************************.
* Table SM6.
*********************************.

* All 5 items.
count nmissing = ENABLE1 ENABLE2 ENABLE3 ENABLE4 ENABLE5 (missing).
do if nmissing=0.
count all5s = ENABLE1 ENABLE2 ENABLE3 ENABLE4 ENABLE5 (5).
count all45s = ENABLE1 ENABLE2 ENABLE3 ENABLE4 ENABLE5 (4,5).
count all12s = ENABLE1 ENABLE2 ENABLE3 ENABLE4 ENABLE5 (1,2).
count all1s = ENABLE1 ENABLE2 ENABLE3 ENABLE4 ENABLE5 (1).
end if.
exe.
freq   all5s all45s all12s all1s .

* Final 3 items.
count nmissing_3item = ENABLE1  ENABLE3  ENABLE5 (missing).
do if nmissing_3item=0.
count all5s_3item = ENABLE1  ENABLE3  ENABLE5(5).
count all45s_3item = ENABLE1  ENABLE3  ENABLE5(4,5).
count all12s_3item = ENABLE1  ENABLE3  ENABLE5 (1,2).
count all1s_3item = ENABLE1  ENABLE3  ENABLE5 (1).
end if.
exe.
freq   all5s_3item all45s_3item all12s_3item all1s_3item .

*********************************.
* Table SM8.
*********************************.

descriptives bjw1 bjw2 sjt1 sjt2 .
RELIABILITY  /VARIABLES= bjw1 bjw2 sjt1 sjt2
  /SCALE('ALL VARIABLES') ALL   /MODEL=ALPHA   /STATISTICS=HOTELLING   /SUMMARY=TOTAL. 
FACTOR   /VARIABLES bjw1 bjw2 sjt1 sjt2 
   /MISSING LISTWISE 
  /ANALYSIS bjw1 bjw2 sjt1 sjt2
  /PRINT UNIVARIATE INITIAL KMO EXTRACTION ROTATION
  /CRITERIA MINEIGEN(1) ITERATE(25)   /EXTRACTION PAF   /CRITERIA ITERATE(25)
  /ROTATION VARIMAX   /METHOD=CORRELATION.


* End of analysis.






















